


WEIRD INVADERS



CLASSIFICATION: Shoot-up game

Guide your ship through this strange galaxy avoiding and shooting the monsters. You are doomed to run out of fuel and die, so you may as well shoot your way out.

Use 'A' to move up, 'Z' to move down, and 'space' to fire.

PROGRAM

Variables

HEIGHT	Row of ship
SCORE	Player's score
ALIENX(4), ALIENY(4)	Coordinates of aliens (col, row)
FUEL	Amount of fuel left
Y	Vertical coordinates of ship
T	Temp

Program Structure

Lines	Function/Activity
50 — 120	Initialise
160 — 270	Main loop
310 — 380	Firing routine
420 — 460	Game over

WEIRD INVADERS

Initialise game

```

50  SYMBOL 240, 0, 9, 4, 81, 2, 40, 3, 0 : SYMBOL 241, 0, 2
    24, 127, 248, 120, 255, 127, 255 : SYMBOL 242, 0, 0, 224
    , 16, 8, 255, 255, 252 : SYMBOL 243, 60, 90, 126, 102, 6
    0, 66, 129, 129
60  HEIGHT = 12
70  SCORE = 0
80  DIM ALIENX(4), ALIENY(4) : FOR T = 1 TO 4 : ALIENX(T) =
    CINT( RND(1) * 20 + 18 ) : ALIENY(T) = CINT( RND(1) * 2
    0 + 1) : NEXT T
90  MODE 1 : BORDER 0 : PAPER 0 : PEN 2 : INK 0, 0 : INK 1,
    18 : INK 2, 24 : INK 3, 7, 24 : SPEED INK 10, 5 : ENV 1
    , 7, -1, 10, 8, -1, 40
100 LOCATE 1, 25 : PRINT "SCORE.....FUEL.
    ....."
110 FUEL = 50
120 LOCATE 1, HEIGHT : PEN 3 : PRINT CHR$(240); : PEN 1 : P
    RINT CHR$(241); CHR$(242) : PEN 2

```

Main program loop

```

130 :
160 FOR T = 1 TO 4
170 LOCATE ALIENX(T), ALIENY(T) : PRINT "▲"
180 ALIENX(T) = ALIENX(T) - 1 : IF ALIENX(T) < 3 THEN IF AL
    IENY(T) = HEIGHT THEN GOTO 420 ELSE ALIENX(T) = 39 : ALI
    ENY(T) = CINT( RND(1) * 20 + 1)
190 LOCATE ALIENX(T), ALIENY(T) : PRINT CHR$(243)
200 NEXT T
210 IF INKEY(69) > -1 THEN LOCATE 1, HEIGHT : PRINT "▲▲▲" :
    HEIGHT = HEIGHT - 1 : LOCATE 1, HEIGHT : PEN 3 : PRINT
    CHR$(240); : PEN 1 : PRINT CHR$(241); CHR$(242) : PEN 2
220 IF HEIGHT < 2 THEN LOCATE 1, 1 : PRINT "▲▲▲" : HEIGHT =
    23 : LOCATE 1, 23 : PEN 3 : PRINT CHR$(240); : PEN 1 :
    PRINT CHR$(241); CHR$(242) : PEN 2
230 IF INKEY(71) > -1 THEN LOCATE 1, HEIGHT : PRINT "▲▲▲" :
    HEIGHT = HEIGHT + 1 : LOCATE 1, HEIGHT : PEN 3 : PRINT
    CHR$(240); : PEN 1 : PRINT CHR$(241); CHR$(242) : PEN 2
240 IF HEIGHT > 23 THEN LOCATE 1, 24 : PRINT "▲▲▲" : HEIGHT
    = 2 : LOCATE 1, 2 : PEN 3 : PRINT CHR$(240); : PEN 1 :
    PRINT CHR$(241); CHR$(242) : PEN 2
250 IF INKEY(47) > -1 THEN GOSUB 310
260 FUEL = FUEL - 1 : LOCATE 35, 25 : PRINT FUEL : IF FUEL
    <= 1 THEN GOTO 420
270 GOTO 160

```

Firing routine, score update and fuel monitor

```

280 :
310 Y = ( 25 - HEIGHT ) * 16 + 4
320 MOVE 48, Y : DRAW 640, Y, 3
330 SOUND 129, 100, 50, 15, 1, 1, 1

```

```
340 FOR T = 1 TO 4
350 IF HEIGHT = ALIENY(T) THEN SCORE = SCORE + T * 10 : LOC
ATE 20, 25 : PRINT SCORE : FUEL = FUEL + 5 : LOCATE ALIE
NX(T), ALIENY(T) : PRINT "▲" : SOUND 2, 1000, 50, 15, 1,
1, 1 : ALIENX(T) = CINT( RND(1) * 20 + 18 ) : ALIENY(T)
= CINT( RND(1) * 20 + 1 )
360 NEXT T
370 MOVE 48, Y : DRAW 640, Y, 0
380 FUEL = FUEL - 1 : RETURN
```

Game over

```
390 :
420 SOUND 2, 200, 100, 1, 1, 1
430 LOCATE 1, HEIGHT : PRINT "▲▲"
440 LOCATE 20, 25 : PEN 3 : PRINT SCORE
450 FOR T = 1 TO 3000 : NEXT T
460 RUN
```

ChexSum Tables

50 = 7390	200 = 433	350 = 14934
60 = 890	210 = 8320	360 = 433
70 = 774	220 = 6294	370 = 1222
80 = 7945	230 = 8289	380 = 1583
90 = 3763	240 = 6293	390 = 0
100 = 2662	250 = 1684	420 = 867
110 = 755	260 = 3981	430 = 1198
120 = 3963	270 = 382	440 = 1322
130 = 0	280 = 0	450 = 1624
160 = 987	310 = 2008	460 = 202
170 = 2362	320 = 1225	
180 = 10365	330 = 939	
190 = 2869	340 = 1019	TOTAL = 108977